carga en movimiento

definiciones de velocidad

productos punto

```
αpn[θ_, ϕ_, αx_, αy_, αz_] := αx Cos[θ] Cos[ϕ] + αy Sin[θ] Cos[ϕ] + αz Sin[ϕ]

αpnxyz[x_, y_, z_, αx_, αy_, αz_] := αx x + αy y + αz z

βpn[θ_, ϕ_, βx_, βy_, βz_] := βx Cos[θ] Cos[ϕ] + βy Sin[θ] Cos[ϕ] + βz Sin[ϕ]

βpnxyz[x_, y_, z_, βx_, βy_, βz_] := βx x + βy y + βz z

αpβ[αx_, αy_, αz_, βx_, βy_, βz_] := αx βx + αy βy + αz βz

Evel[θ_, ϕ_, βx_, βy_, βz_] :=
  ((1 - (βx^2 + βy^2 + βz^2)) {Cos[θ] Cos[ϕ] - βx , Sin[θ] Cos[ϕ] - βy , Sin[ϕ] - βz}) /
   (1 - βpn[θ, ϕ, βx, βy, βz])^3

Evelxyz[x_, y_, z_, βx_, βy_, βz_] :=
  ((1 - (βx^2 + βy^2 + βz^2)) {x - βx , y - βy , z - βz}) / (1 - βpnxyz[x, y, z, βx, βy, βz])^3

Evelmag[θ_, ϕ_, βx_, βy_, βz_] :=
  ((1 - (βx^2 + βy^2 + βz^2)) √((Cos[θ] Cos[ϕ] - βx)^2 + (Sin[θ] Cos[ϕ] - βy)^2 + (Sin[ϕ] - βz)^2)) /
   (1 - βpn[θ, ϕ, βx, βy, βz])^3
```

definiciones de aceleración

```
num[θ_, ϕ_, αx_, αy_, αz_, βx_, βy_, βz_] := (
   αpn[θ, ϕ, αx, αy, αz]^2 * (1 - 2 βpn[θ, ϕ, βx, βy, βz] + (βx^2 + βy^2 + βz^2))
    - 2 * (1 - βpn[θ, ϕ, βx, βy, βz]) *
     (αpn[θ, ϕ, αx, αy, αz]) * (αpn[θ, ϕ, αx, αy, αz] - αpβ[αx, αy, αz, βx, βy, βz])
    + (1 - βpn[θ, ϕ, βx, βy, βz])^2 * (αx^2 + αy^2 + αz^2)
  )

Eacel[θ_, ϕ_, αx_, αy_: 0, αz_: 0, βx_: 0, βy_: 0, βz_: 0] :=
  (αpn[θ, ϕ, αx, αy, αz] {Cos[θ] Cos[ϕ] - βx , Sin[θ] Cos[ϕ] - βy , Sin[ϕ] - βz} -
     (1 - βpn[θ, ϕ, βx, βy, βz]) {αx, αy, αz}) / (1 - βpn[θ, ϕ, βx, βy, βz])^3

Eacelxyz[x_, y_, z_, αx_, αy_: 0, αz_: 0, βx_: 0, βy_: 0, βz_: 0] :=
  (αpnxyz[x, y, z, αx, αy, αz] {x - βx , y - βy , z - βz} -
     (1 - βpnxyz[x, y, z, βx, βy, βz]) {αx, αy, αz}) / (1 - βpnxyz[x, y, z, βx, βy, βz])^3

Eacelmag[θ_, ϕ_, αx_, αy_: 0, αz_: 0, βx_: 0, βy_: 0, βz_: 0] := √(num[θ, ϕ, αx, αy, αz, βx, βy, βz]) / (1 - βpn[θ, ϕ, βx, βy, βz])^3

Etot[θ_, ϕ_, αx_, αy_, αz_, βx_, βy_, βz_, R_: 1] :=
  1/R^2 Evel[θ, ϕ, βx, βy, βz] + 1/R Eacel[θ, ϕ, αx, αy, αz, βx, βy, βz]

Etotmag[θ_, ϕ_, αx_, αy_, αz_, βx_, βy_, βz_, R_: 1] :=
  1/R^2 Evelmag[θ, ϕ, βx, βy, βz] + 1/R Eacelmag[θ, ϕ, αx, αy, αz, βx, βy, βz]
```

términos de velocidad

```
Simplify[Evelmag[θ, 0, 0, βy, 0]]
```

$$\frac{\left(-1+\beta y^2\right)\sqrt{1+\beta y^2-2\,\beta y\,\mathrm{Sin}[\theta]}}{\left(-1+\beta y\,\mathrm{Sin}[\theta]\right)^3}$$

```
Plot[{Evelmag[θ, 0, 0, 0, 0], Evelmag[θ, 0, 0, 0.4`, 0],
  Evelmag[θ, 0, 0, 0.7`, 0], Evelmag[θ, 0, 0, 0.9`, 0]}, {θ, -π, π}, AxesLabel → {θ, βy},
 PlotRange → All, PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```




```
polplotvels =
  PolarPlot[{Evelmag[θ, 0, 0, 0, 0], Evelmag[θ, 0, 0, 0.4, 0], Evelmag[θ, 0, 0, 0.7, 0],
    Evelmag[θ, 0, 0, 0.9, 0]}, {θ, -π, π}, AxesLabel → {x, βy}, PlotRange → All,
   PlotStyle → {{Hue[0.1]}, {Hue[0.3]}, {Hue[0.6]}, {Hue[0.8]}}, DisplayFunction → Identity];

polplotvelsdet = PolarPlot[{Evelmag[θ, 0, 0, 0, 0], Evelmag[θ, 0, 0, 0.4, 0],
    Evelmag[θ, 0, 0, 0.7, 0], Evelmag[θ, 0, 0, 0.9, 0]}, {θ, -π, π},
   AxesLabel → {x, βy}, PlotStyle → {{Hue[0.1]}, {Hue[0.3]}, {Hue[0.6]}, {Hue[0.8]}},
   PlotRange → {{-2.5, 2.5}, {-1, 2.5}}, DisplayFunction → Identity];

Show[GraphicsRow[{polplotvels, polplotvelsdet}], DisplayFunction → $DisplayFunction]
```




```
plotpolarvel0 = PolarPlot[{Evelmag[θ, 0, 0, 0, 0]}, {θ, -π, π}, AxesLabel → {x, βy},
   PlotRange → All, PlotStyle → {{Hue[0.1]}, {Hue[0.3]}, {Hue[0.6]}, {Hue[0.8]}}];

VectorPlot[{-1 - θ^2 + ϕ, 1 + θ - ϕ^2}, {θ, -3, 3}, {ϕ, -3, 3},
  VectorPoints → RandomReal[{-3, 3}, {300, 2}], VectorScale → Large];

points2d = Table[{Cos[θ], Sin[θ], 0}, {θ, 0, 2 Pi, Pi / 16}];
```

```
VectorPlot3D[ Evelxyz[x, y, z, 0, 0, 0], {x, -1, 1 }, {y, -1, 1}, {z, -.1, .1},
 VectorPoints → points2d, VectorScale → Medium, BoxRatios → Automatic]
```




```
points3d = Flatten[
   Table[{Cos[θ] Cos[ϕ], Sin[θ] Cos[ϕ], Sin[ϕ]}, {θ, 0, 2 Pi, Pi / 6}, {ϕ, 0, 2 Pi, Pi / 8}], 1];
```

```
VectorPlot3D[ Evelxyz[x, y, z, 0, 0, 0], {x, -1, 1 }, {y, -1, 1}, {z, 0, 1},
 VectorPoints → points3d, VectorScale → Medium, BoxRatios → Automatic]
```




```
plot2dvel0 = VectorPlot3D[ Evelxyz[x, y, z, 0, 0, 0], {x, -1, 1 }, {y, -1, 1},
   {z, -.1, .1}, VectorPoints → points2d, VectorScale → Medium, BoxRatios → Automatic];
```

```
plot3dvel0 = VectorPlot3D[ Evelxyz[x, y, z, 0, 0, 0], {x, -1, 1 }, {y, -1, 1},
   {z, 0, 1}, VectorPoints → points3d, VectorScale → Medium, BoxRatios → Automatic];
```

```
{plotpolarvel0, plot2dvel0, plot3dvel0}
```


β_y
1.0
0.5
−0.5
−1.0
−1.0
−0.5
0.5
1.0
x'


0.2
0.1
0.0
-0.1
-0.2
−1.0
−0.5
0.0
0.5
1.0
0.5
0.0
−0.5
−1.0




```
plot2dvel7 = VectorPlot3D[ Evelxyz[x, y, z, 0, 0.4, 0], {x, -1, 1 }, {y, -1, 1},
  {z, -.1, .1}, VectorPoints → points2d, VectorScale → {.2, Scaled[.4], Automatic},
  BoxRatios → Automatic, PlotRange → {{-1.5, 1.5}, {-1.5, 1.5}, {-.1, .1}}]
```



```
VectorPlot3D[ Evelxyz[x, y, z, 0, 0.4, 0], {x, -1, 1 }, {y, -1, 1},
 {z, -.1, .1}, VectorPoints → points2d, VectorScale → Automatic,
 BoxRatios → Automatic, PlotRange → {{-1.5, 1.5}, {-1.5, 1.5}, {-.1, .1}}]
```




```
plotpolarvel7 = PolarPlot[{Evelmag[θ, 0, 0, 0.4, 0]}, {θ, -π, π},
  AxesLabel → {x, βy}, PlotRange → All, PlotStyle → {{Hue[0.6]}, {Hue[0.8]}}]
```

General::ivar : 3.3877070526190334` is not a valid variable. >>



```
{plotpolarvel7, plot2dvel7}
```

{ 



}

$$\texttt{PolarPlot}\left[\frac{1-0.64`}{\sqrt{\left(1-0.64`\,\texttt{Sin}[\theta]^2\right)^3}},\ \{\theta,\ -\pi,\ \pi\}\right]$$



aceleración

- **aceleración para velocidad cero.**

```
Simplify[Eacelmag[θ, 0, αx]]
```

$$\sqrt{\texttt{Sin}[\theta]^2\,\alpha_x^2}$$

```
PolarPlot[{Eacelmag[θ, 0, 1], Eacelmag[θ, 0, 0.5`], Eacelmag[θ, 0, 0.1`]}, {θ, -π, π},
 AxesLabel → {α, y}, PlotRange → All, PlotStyle → {{Hue[0.5`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



- **aceleración para velocidad colineal arbitraria.**

```
PolarPlot[{Eacelmag[θ, 0, 1, 0, 0, 0],
  Eacelmag[θ, 0, 1, 0, 0, 0.5`], Eacelmag[θ, 0, 1, 0, 0, 0.7`]}, {θ, -π, π},
 AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\),\!\(\*SubscriptBox[\(β\), \(x\)]\)", y},
 PlotRange → All, PlotStyle → {{Hue[0.5`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```

```
General::ivar : 3.3877070526190334` is not a valid variable. >>
```




```
Simplify[Eacelmag[θ, 0, αx, 0, 0, βx]]
```

$$-\frac{\sqrt{\text{Sin}[\theta]^2\,\alpha_x^2}}{(-1+\text{Cos}[\theta]\,\beta_x)^3}$$

aceleración para velocidad perpendicular.

```
Simplify[Eacelmag[θ, 0, αx, 0, 0, 0, βy, 0]]
```

$$-\frac{\sqrt{\alpha x^2\ (\beta y - \text{Sin}[\theta])^2}}{(-1 + \beta y\ \text{Sin}[\theta])^3}$$

```
PolarPlot[{Eacelmag[θ, 0, 1, 0, 0, 0], Eacelmag[θ, 0, 1, 0, 0, 0, 0.5`],
  Eacelmag[θ, 0, 1, 0, 0, 0, 0.6`], Eacelmag[θ, 0, 1, 0, 0, 0, 0.65`]}, {θ, -π, π},
 AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\)", βy}, PlotRange → All,
 PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



- **aceleración para velocidad colineal y perpendicular.**

```
PolarPlot[{Eacelmag[θ, 0, 1, 0, 0, 0], Eacelmag[θ, 0, 1, 0, 0, 0.5`],
  Eacelmag[θ, 0, 1, 0, 0, 0, 0.5`], Eacelmag[θ, 0, 1, 0, 0, 0.5`, 0.5`]}, {θ, -π, π},
 AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\)", βy}, PlotRange → All,
 PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



términos de aceleración
velocidad cero (naranja)
velocidad colineal (verde)
velocidad ortogonal a aceleración (azul)
velocidad en ambas direcciones (morado)

```
plotpolaracel1 = PolarPlot[{Eacelmag[θ, 0, 1]}, {θ, -π, π},
   AxesLabel → {x, βy}, PlotRange → All, PlotStyle → {{Hue[0.6]}, {Hue[0.8]}}];
```

```
General::ivar : 3.3877070526190334` is not a valid variable. >>
```

```
plot2dacel1 = VectorPlot3D[ Eacelxyz[x, y, z, 1, 0, 0], {x, -1, 1 }, {y, -1, 1},
  {z, -.1, .1}, VectorPoints → points2d, VectorScale → Medium, BoxRatios → Automatic]
```



```
{plotpolaracel1, plot2dacel1}
```

{





}






```
plotpolaracel1vel5col = PolarPlot[{Eacelmag[θ, 0, 1, 0, 0, 0.5]}, {θ, -π, π},
   AxesLabel → {x, βy}, PlotRange → All, PlotStyle → {{Hue[0.6]}, {Hue[0.8]}}];
```

General::ivar : 3.3877070526190334` is not a valid variable. >>

```
plot2dacel1vel5col =
 VectorPlot3D[ Eacelxyz[x, y, z, 1, 0, 0, .5], {x, -1.3, 1.3 }, {y, -1.3, 1.3},
  {z, -.1, .1}, VectorPoints → points2d, VectorScale → Automatic, BoxRatios → Automatic]
```




```
{plotpolaracel1vel5col, plot2dacel1vel5col}
```

{





}






```
plotpolaracel1vel5per =
  PolarPlot[{Eacelmag[θ, 0, 1, 0, 0, 0, 0.5]}, {θ, -π, π}, AxesLabel → {x, βy},
   PlotRange → All, PlotStyle → {{Hue[0.6]}, {Hue[0.8]}}, DisplayFunction → Identity];
```

General::ivar : 3.3877070526190334` is not a valid variable. >>

```
plot2dacel1vel5per =
 VectorPlot3D[ Eacelxyz[x, y, z, 1, 0, 0, 0, .5], {x, -1.3, 1.3 }, {y, -1.3, 1.3},
  {z, -.1, .1}, VectorPoints → points2d, VectorScale → Automatic, BoxRatios → Automatic]
```



```
{plotpolaracel1vel5per, plot2dacel1vel5per}
```

{





}





contribuciones de términos de velocidad y aceleración

```
PolarPlot[{Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 1], Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 3],
  Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 5], Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 10]}, {θ, -π, π},
 AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\)", y}, PlotRange → All,
 PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



para velocidad cero y aceleración en x, el campo estático contribuye para distancias cercanas a la carga.

```
PolarPlot[{Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 10], Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 30],
  Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 50], Etotmag[θ, 0, 1, 0, 0, 0, 0, 0, 100]},
 {θ, -π, π}, AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\)", y}, PlotRange → All,
 PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



para velocidad cero y aceleración en x, el campo estático ya no contribuye a mayores distancias.

```
PolarPlot[{Etotmag[θ, 0, 1, 0, 0, 0.5`, 0, 0, 1], Etotmag[θ, 0, 1, 0, 0, 0.5`, 0, 0, 3],
  Etotmag[θ, 0, 1, 0, 0, 0.5`, 0, 0, 5], Etotmag[θ, 0, 1, 0, 0, 0.5`, 0, 0, 10]},
 {θ, -π, π}, AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\)", y}, PlotRange → All,
 PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



velocidad colineal, el campo de velocidad y de aceleración ambos contribuyen a distancias cercanas.
Para mayores distancias con velocidad colineal, el campo de aceleración contribuye pero el campo de velocidades es ya pequeño.

```
PolarPlot[{Etotmag[θ, 0, 1, 0, 0, 0, 0.5`, 0, 1], Etotmag[θ, 0, 1, 0, 0, 0, 0.5`, 0, 3],
  Etotmag[θ, 0, 1, 0, 0, 0, 0.5`, 0, 5], Etotmag[θ, 0, 1, 0, 0, 0, 0.5`, 0, 10]},
 {θ, -π, π}, AxesLabel → {"\!\(\*SubscriptBox[\(α\), \(x\)]\)", βy}, PlotRange → All,
 PlotStyle → {{Hue[0.1`]}, {Hue[0.3`]}, {Hue[0.6`]}, {Hue[0.8`]}}]
```



aceleración 3D en general